' Retrieve Class Info - © 1993-2002 Aiman Hanna

' This program explores how files can be read.

' Key Points:

'       1) Files

'       2) The "ReadLine" function of TextStream

'       3) The "AtEndOfStream" function of TextStream

' Very Important:

'    Notice that you must include the "Microsoft Scripting Runtime". You

'    include this by selecting the "Project" tab ( References, then click

'    on (select) the "Microsoft Scripting Runtime" from the list of available references.

'    If you do not include this reference, you will NOT be able to use any of the

'    FileSystemObject, File, or TextStream classes and hence none of the functionality

'    that these classes provide.

Option Explicit

Dim fso As New FileSystemObject

Dim f1 As File

Dim txtstrm As TextStream

Private Sub Form_Load()

   ' This sub will be called automatically (and only once)

   'when the form is loaded at run-time (when the program starts)

   If fso.FileExists("c:\comp201\c201_info.dat") = False Then

        Call MsgBox("Input File does not exist. Program will termiante.")

       ' File does not exist. Can not read anything, so terminate the program

       End

   End If

   Set f1 = fso.GetFile("c:\comp201\c201_info.dat")

   ' Open a text stream for writing to the file

   Set txtstrm = f1.OpenAsTextStream(ForReading)

   ' Display path in lblFileName

   lblFileName.Caption = "All information will be read from : " & f1.Path

End Sub

Private Sub cmdDisplayClassInfo_Click()

    Dim s As String, tempstr As String

    Dim id As String

    Dim spacepos As Integer

    ' Make sure that you are not at the end of the stream (end of the file).

    ' This is needed since this "Display Class Info" button may be clicked

    ' more than once.

    If txtstrm.AtEndOfStream = True Then

       Call MsgBox("Class info already shown.", vbOKOnly, "Information Message")

    Else    ' Read the file and display its information on the list

    ' Read the file line by line and display each line on the list.

    ' Add each id to the combo box as well

        Do

            s = txtstrm.ReadLine

            lstClassInfo.AddItem s

            ' extract the id from the line

            spacepos = InStr(s, " ")

            tempstr = Mid$(s, spacepos)

            ' id is the first part (from left) of the string

            id = Left$(s, Len(s) - Len(tempstr))

            cmbIDList.AddItem id

        Loop While txtstrm.AtEndOfStream = False

    End If

End Sub

Private Sub form_terminate()

    txtstrm.Close

End Sub

Private Sub cmdExit_Click()

    End

End Sub

